

&**PURPOSE**

This keyword is used to concatenate two previously created strings.

DESCRIPTION

Strings are created with either the READ STRING command or the LET STRING command. The READ STRING can read strings either from the terminal or from a file. Another distinction is that READ STRING preserves the case of the string while LET STRING converts all characters to upper case.

The & keyword is restricted to the LET STRING case.

SYNTAX

None

EXAMPLES

```
LET STRING FNAME = CALIB.DAT
LET STRING PATH = /usr/nist/heckert/datafiles/
LET STRING NAME = ^PATH&^FNAME
READ ^NAME X1 X2 X3
```

NOTE 1

When concatenating strings, you need to use the substitution character (^) in front of the strings. For example, given

```
LET STRING A = CUBE01
LET STRING B = .DAT
```

then the following

```
LET STRING C = ^A&^B
```

generates the string CUBE01.DAT. However,

```
LET STRING D = A&B
```

generates the string A&B.

NOTE 2

The following commands

```
LET STRING A = CUBE01
LET STRING C = ^A&.DAT
```

generate the string CUBE01.DAT as expected. However, the following command

```
LET STRING A = CUBE01&.DAT
```

generates the string CUBE01&.DAT instead of CUBE01.DAT.

NOTE 3

DATAPLOT does not at this time support commands for extracting a subset of a string or for returning the number of characters in an already existing string.

NOTE 4

More than one concatenation character can appear on a line.

NOTE 5

The & and ^ keywords can be effectively used with the LOOP command for reading and analyzing similarly structured files. The program example below is used to analyze 100 files named CALIB1.DAT, CALIB2.DAT, ..., CALIB100.DAT. These files are assumed to all exist, have 5 header lines, and 2 columns of data (the number of rows can vary). Each of these will be analyzed with a DATAPLOT macro.

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

LET STRING	=	Define a string.
READ STRING	=	Read a string.
^	=	Substitute the value of a string or a parameter into a command.

APPLICATIONS

Concatenating strings, useful in macros

IMPLEMENTATION DATE

89/2

PROGRAM

```
LET STRING S1 = CALIB
LET STRING S3 = .DAT
SKIP 5
LOOP FOR K = 1 1 100
  DELETE X Y
  LET STRING S2 = ^K
  LET STRING S4 = ^S1&^S2&^S3
  READ ^S4 X Y
  CALL ANALYSIS.DP
END OF LOOP
```